

# Integracja CTI z Microsoft Access

Instalacja, Konfiguracja i Obsługa

## Spis treści

1 Przygotowanie.....	3
2 Cel.....	5
3 Powiązanie.....	5
4 Testowanie połączeń wychodzących.....	13
5 Testowanie identyfikacji wywołań.....	15
6 Dalsze kroki .....	18

# 1 Przygotowanie

Na komputerze, na którym będzie przeprowadzona integracja CTI z bazą danych Nordwind, musi być zainstalowany działający klient OSITRON CTI 4 (przy czym również instalacja serwera zawiera także klienta) jak również Microsoft Access 2010.

Baza danych Nordwind jest to przykładowa baza danych dostarczana w ramach testów.


Bazę danych Nordwind można otworzyć w Microsoft Access 2010. Prowadzą do tego poniższe kroki:

- Uruchomić Microsoft Access 2010.
- Wybrać menu *Plik -> Nowe*.
- Kliknąć w obszarze *Dostępne szablony* na *Przykładowe szablony*.
- Teraz w przykładowych szablonych kliknąć na wpisie *Nordwind*.
- Po pobraniu i otwarciu bazy danych Nordwind, wymagane jest potwierdzenie żółtego ostrzeżenia o zabezpieczeniach w górnym obszarze, przez użycie przycisku *Włącz zawartość*:



- Teraz można używać bazę danych Nordwind.

Należy się także upewnić, że w Centrum Telefonicznym została wybrana właściwa linia TAPI:



Poza tym okno stanu OSITRON CTI musi sygnalizować właściwe numery telefonów rozmówców przy połączeniach przychodzących i wychodzących.

W tym celu należy prawidłowo, w sposób podany poniżej, skonfigurować lokalizację Windows, która znajduje się w aplecie *Opcje telefonu i modemu* w *Panelu Sterowania* Windows:

- Musi istnieć *Domyślna lokalizacja* z ustawieniami domyślnymi. Domyślna lokalizacja jest wyróżniona na liście lokalizacji pogrubioną czcionką.
- W tej lokalizacji musi być wpisany *Numer kierunkowy* dla strefy telekomunikacyjnej, bez poprzedzających zer.
- Dodatkowo należy podać prefiksy dostępu do *linii zewnętrznej*, które są wykorzystywane przez daną centralę telefoniczną.

Ważne jest, aby sygnalizowany numer telefonu był wyświetlany tak, jak jest wybierany z telefonu bez centrali telefonicznej. Na przykład, sygnalizowany numer telefonu 223791730 jest prawidłowy, podczas gdy numer 0223791730 lub 00223791730 nie jest prawidłowy.

Numery telefonów dla rozmów lokalnych lub międzynarodowych są określone wyłącznie przez dostawcę telefonii (sterownik TAPI lub centrala telefoniczna). One decydują, czy sygnalizowany jest odpowiedni numer kierunkowy czy nie. Jeśli numer sygnalizowany jest w sposób nieprawidłowy i nie zawiera numeru kierunkowego, to należy się skontaktować z administratorem ds. telefonii.

Dodatkowo numery telefonów przyporządkowane do rekordów bazy danych muszą zawierać prawidłowe numery telefonów lub być podane w postaci kanonicznej.

Właściwy numer telefonu w postaci kanonicznej ma następujący wygląd:  
+<NumerKraju><NumerStrefyBezZera><NumerPrzyłącza>

Przykład:  
*+48 (22) 3791730*

## 2 Cel

Ta instrukcja opisuje rozbudowę formularzy z danymi klientów przykładowej bazy danych Nordwind z Microsoft Access 2010 o podstawową funkcjonalność programu OSITRON CTI 4.

Po pierwsze powinna być zapewniona możliwość zadzwonienia za pomocą przycisku znajdującego się obok pola z numerem telefonu.

Po drugie, na podstawie zdarzeń sygnalizujących połączenia przy- i wychodzące, powinien zostać wyświetlony odpowiedni rekord z danymi klienta.


## 3 Powiązanie

Po otwarciu bazy danych Nordwind w Microsoft Access, jak opisano w rozdziale [1. Przygotowanie](#) i zalogowaniu się jako dowolny użytkownik przykładowy, można rozpocząć przeprowadzenie powiązania bazy danych Nordwind z OSITRON CTI.


Należy otworzyć Edytor Visual Basic za pomocą kombinacji klawiszy *Alt+F11* lub przez polecenie menu *Tworzenie* na wstążce *Visual Basic* w grupie *Makra i Kod*:



W edytorze Visual Basic należy użyć polecenia menu *Tools* i *References*:




W oknie *References - Nordwind* należy uaktywnić odnośnik do OSICTIControlCenter 3.0 Typbibliothek:



Należy nacisnąć na przycisk **OK**, aby zatwierdzić odnośnik.

Program *Microsoft Visual Basic for Applications - Nordwind* jest nadal otwarty, należy nacisnąć w oknie *Projektu* prawym klawiszem myszy na *Class modules* i w menu kontekstowym na *Insert > i* dalej *Class module*:



Zostanie otwarty moduł o nazwie *Nordwind - Class1 (Code)*. Należy wpisać w edytorze poniższy kod:

*Option Explicit*

```
Public WithEvents CtiAppClient As OSICTIControlCenter.OsiCtiApp
Public WithEvents CtiCallClient As OSICTIControlCenter.OSICTICall
```

*' Beim Initialisieren OsiCtiApp instanziiieren.*

```
Private Sub Class_Initialize()
```

```
    If CtiAppClient Is Nothing Then
```

```
        Set CtiAppClient = New OSICTIControlCenter.OsiCtiApp
```

```
    End If
```

```
End Sub
```

*' Beim Beenden Speicher freisetzen und Verbindung zu CTI beenden.*

```
Private Sub Class_Terminate()
```

```
    Set CtiCallClient = Nothing
```

```
    Set CtiAppClient = Nothing
```

```
End Sub
```

*' Eine ausgehende Wahl für die angegebene Rufnummer durchführen.*

```
Public Sub MakeCall(CallNumber As String)
```

```
    On Error GoTo errorMsg
```

```
    Dim NewCall As OSICTIControlCenter.OSICTICall
```

```
    Set NewCall = CtiAppClient.CreateCall
```

```
    NewCall.DialAddress = CallNumber
```

```
    NewCall.Dial
```

```
    Set NewCall = Nothing
```

```
Exit Sub
```

```
errorMsg:
```

```
    MsgBox "Beim Aufruf der Wählfunktion ist folgender Fehler  
aufgetreten: " & Chr(13) & Err.Number & " / " & Err.Description
```

```
End Sub
```

*' CtiCallClient einen neu signalisierten neuen Anruf zuordnen.*

```
Private Sub CtiAppClient_NewCallEvent(ByVal piApp As
```

```
OSICTIControlCenter.IOSICTIApp, ByVal piCall As
```

```
OSICTIControlCenter.OSICTICall)
```

```
    Set CtiCallClient = piCall
```

```
End Sub
```

*' Anhand des Ereignisses CalledAddressEvent eine Kundensuche durchführen.*

```
Private Sub CtiCallClient_CalledAddressEvent(ByVal piCall As
```

```
OSICTIControlCenter.OSICTICall, ByVal bstrCalledAddress As String)
```

```
    Dim sRemoteCallNumber As String
```

```
If piCall.Direction = OSICTICallDirectionIncoming Then
    sRemoteCallNumber = piCall.CallingAddress

Else
    sRemoteCallNumber = piCall.CalledAddress

End If

FindCustomerByCallNumber sRemoteCallNumber

End Sub

' Anhand des Ereignisses CallingAddressEvent eine Kundensuche durchführen.
Private Sub CtiCallClient_CallingAddressEvent(ByVal piCall As
OSICTIControlCenter.IOSICTICall, ByVal bstrCallingAddress As String)

    Dim sRemoteCallNumber As String

    If piCall.Direction = OSICTICallDirectionIncoming Then
        sRemoteCallNumber = piCall.CallingAddress

    Else
        sRemoteCallNumber = piCall.DialAddress

    End If

    FindCustomerByCallNumber sRemoteCallNumber

End Sub

' Eine Kundensuche auf Basis der signalisierten Rufnummer der Gegenstelle
' durchführen und den identifizieren Datensatz in den Kundendetails
' anzeigen.
Private Sub FindCustomerByCallNumber(CallNumber As String)

    Dim dbs As Database
    Set dbs = CurrentDb

    Dim rst As Recordset
    Set rst = Form_Kundendetails.RecordsetClone

    If rst.RecordCount < 1 Then
        rst.Close

    Else
        rst.FindFirst "[Telefon (geschäftlich)] = '" & CallNumber & _
            "' OR [Mobiltelefon] = '" & CallNumber & "'"
        If Not rst.NoMatch Then
            Form_Kundendetails.Bookmark = rst.Bookmark

        End If

    End If

    rst.Close
    Set rst = Nothing

End Sub
```

Zapisać i zamknąć okno, zawierające wpisany kod, pod nazwą *OsiCtiCallClient*.

Jak już opisano w rozdziale [2\\_Cel](#) formularz danych klienta zostanie wyposażony w funkcjonalność CTI.


Należy rozszerzyć z lewej strony *Obszar nawigacji* przez naciśnięcie na symbol >>:




Dzięki temu zostanie wyświetlona zawartość nawigacji. W rubryce *Klienci i zamówienia* widoczne są „często używane” odnośniki do formularzy, zapytania lub inne obiekty Access.

Na tej liście widać także odnośnik *Szczegółowe dane klienta*, który przedstawia informacje o kliencie w postaci formularza ze szczegółami i który ma zostać wyposażony w funkcjonalność CTI.

Należy kliknąć prawym klawiszem myszki na wpis *Szczegółowe dane klienta* i następnie w menu kontekstowym wybrać *Widok projektu*:



Teraz otwarty jest *Widok projektu* formularza *Szczegółowe dane klienta*. Na skrzyżowaniu współrzędnych formularza, z lewej strony u góry, należy kliknąć przycisk z kwadratem (1). Następnie w wyświetlonym z prawej strony *Arkuszu właściwości* wybrać zakładkę *Zdarzenie* i dalej na przycisk „...“ na wpisie *Przy otwarciu* (2):



Pod linią procedury zdarzenia *Private Sub Form\_Open(Cancel As Integer)* należy wpisać następujący kod:

```

If OsiCtiClnt Is Nothing Then
    Set OsiCtiClnt = New OsiCtiCallClient
End If

```

To odwołanie zainicjuje moduł klasy *OsiCtiCallClient* przy otwarciu formularza *Szczegółowe dane klienta* i tym samym utworzone zostanie powiązanie z OSITRON CTI.


Pomiędzy predefiniowaną linią *Option Compare Database* i wyżej wspomnianą metodą *Form\_Open* należy wpisać poniższy kod:

```

Private OsiCtiClnt As OsiCtiCallClient

```

W otwartym dokumencie w menu kontekstowym z prawej strony u góry (1) należy teraz wybrać *Close* (2)...



I w wygenerowanej procedurze zdarzenia *Private Sub Form\_Close()* wpisać:


```

Set OsiCtiClnt = Nothing


```

Teraz należy zamknąć edytor Visual Basic-Editor. Powróci *Widok projektu* formularza *Szczegółowe dane klienta*.


Teraz za pomocą menu *Projektowanie* (1) należy utworzyć dwie kontrolki przycisków (2) dla inicjowania dzwonienia obok pól *Telefon służbowy* i *Telefon komórkowy*.



Znaczyć teraz kolejno obydwa dodane kontrolki i ustawić w prawej strony w *arkuszu właściwości* *Obraz*, oraz dla przycisku telefonu służbowego jako *Nazwa MakeTelCall* a dla telefonu komórkowego jako *Nazwa MakeMobileCall*:



Teraz obydwa przyciski zawierające ustawione ikony powinny znajdować się obok pól *Telefon służbowy* i *Telefon komórkowy* i wyglądać mniej więcej jak poniżej:



Teraz należy zaznaczyć przycisk MakeTelCall (1) i w Arkuszu właściwości wybrać zakładkę Zdarzenie i dalej nacisnąć na przycisk „...” obok zdarzenia Przy kliknięciu (2):



W automatycznie utworzonej procedurze *Private Sub MakeTelCall\_Click()* należy wpisać kod...:

```
If Not IsNull(Me.Business_Phone) Then  
    OsiCtiClnt.MakeCall Me.Business_Phone  
  
End If
```

... oraz zachować i zamknąć edytor Visual Basic.

Ten sam proces należy powtórzyć dla przycisku MakeMobileCall, a w automatycznie wygenerowanej procedurze zdarzenia *Private Sub MakeMobileCall\_Click()* wpisać kod...:

```
If Not IsNull(Me.Mobile_Phone) Then  
    OsiCtiClnt.MakeCall Me.Mobile_Phone  
  
End If
```

... i następnie zapisać i zamknąć edytor Visual Basic.

Integracja podstawowych funkcji programu OSITRON CTI w bazie danych Nordwind jest teraz zaimplementowana i może być używana.

Należy zachować wszystkie nie zapisane dokumenty w Microsoft Access i zakończyć program Access. Należy również zakończyć program Centrum Telefoniczne CTI.


## 4 Testowanie połączeń wychodzących

Po przeprowadzeniu integracji CTI można uruchomić bazę danych Nordwind w celu przeprowadzenia testów.


Centrum telefoniczne CTI powinno zostać uruchomione automatycznie wraz z bazą danych dzięki integracji. Należy się upewnić, że w centrum telefonicznym CTI została wybrana poprawna linia TAPI i możliwe jest inicjowanie połączeń wychodzących za pomocą polecenia w menu kontekstowym w Centrum Telefonicznym CTI → *Zadzwoń*.

Dla przykładowego klienta *Jarosław Dyszkiewicz* należy podać jako numer telefonu służbowego rzeczywiście istniejący numer telefonu, na który można fizycznie zadzwonić z aparatu telefonicznego. W naszym przykładzie jest to numer telefonu 223791730.

W tym celu należy otworzyć bazę danych Nordwind, a następnie z poziomu *Nawigatora* formularz *Szczegółowe dane klienta* (1), zaznaczyć klienta *Jarosław Dyszkiewicz* (Nr rekordu 12) (2) i wpisać rzeczywiście istniejący numer telefonu, w naszym przypadku 223791730 (3):




Następnie należy nacisnąć na przycisk, który jest umiejscowiony obok pola *Telefon służbowy*.



Po naciśnięciu na przycisk, powinno zostać zainicjowane zestawianie połączenia wychodzącego, wykorzystujące zaimplementowaną komendę *MakeCall*. W większości wypadków (w zależności od używanej centrali telefonicznej) należy wcześniej podnieść słuchawkę aparatu telefonicznego, aby mogło nastąpić wybranie numeru telefonu.

Jednocześnie z wybraniem numeru telefonu powinno zostać otwarte okno stanu OSITRON CTI. W tym oknie stanu można odczytać, że przeprowadzane jest właśnie wybieranie numeru telefonu:



Własny, wewnętrzny numer telefonu  
Status  
Numer telefonu rozmówcy


## 5 Testowanie identyfikacji wywołań

Dzięki integracji CTI w bazie danych Nordwind jest możliwa identyfikacja połączeń przy- i wychodzących na podstawie numerów telefonu z bazy danych Nordwind.

W tym celu formularz szczegółowych danych klienta zostanie dalej rozbudowany/rozszerzony, żeby podczas rozmowy, na podstawie numeru telefonu rozmówcy, był wyświetlany rekord klienta.

Także w tym wypadku posłużymy się klientem Jarosław Dyszkiewicz, z rzeczywście istniejącym numerem telefonu, w tym przykładzie ponownie numer 223791730.

W tym celu, w otwartej bazie danych Nordwind, w panelu *Nawigatora* należy otworzyć formularz *Szczegółowe dane klienta* (1), wybrać klienta *Jarosław Dyszkiewicz* (rekord 12) (2) i wpisać istniejący numer telefonu, w naszym przypadku 223791730 (3):




Pozostawić formularz otwarty i przeskoczyć do nawigatora rekordów u dołu okienka i tam wybrać dowolny, inny rekord.

## Połączenia wychodzące

Teraz w menu kontekstowym Centrum Telefonicznego CTI należy wybrać polecenie *Zadzwoń* i wprowadzić w okienku wybierania rzeczywisty numer telefonu, w naszym przypadku 223791730.

Teraz w oknie stanu CTI powinny być widoczne odpowiednie informacje, a z drugiej strony powinien zostać wyświetlony rekord z klientem Jarosław Dyszkiewicz:



### Wskazówka:

Warto przy tym zwrócić uwagę na to, że okienko stanu CTI nie wyświetla danych z bazy Nordwind (np. miejscowość *Las Vegas* czy nazwisko *Jarosław Dyszkiewicz*).


Taką funkcjonalność można osiągnąć przez podłączenie książki adresowej CTI. Kilka przykładów podłączenia książek adresowych oferujemy pod adresem <http://www.ositron.de/de/ositron-cti-addins.html> w rubryce *Książka adresow.*

## Połączenia przychodzące

W celu przeprowadzenia teraz testu dla rozmów przychodzących, należy wybrać dowolnego klienta, który nie nazywa się Jarosław Dyszkiewicz.

Teraz należy zadzwonić z numeru który rzeczywiście istnieje, który został przypisany do Jarosława Dyszkiewicza, na linię, która została wybrana w Centrum Telefonicznym CTI.

Po wykonaniu połączenia powinien zacząć dzwonić własny aparat telefoniczny oraz jednocześnie otworzyć się okienko stanu z odpowiednimi informacjami a także otworzyć się rekord z danymi Jarosława Dyszkiewicza:



### Wskazówka:

Warto przy tym zwrócić uwagę na to, że okienko stanu CTI nie wyświetla danych z bazy Nordwind (np. miejscowość Las Vegas czy nazwisko Jarosław Dyszkiewicz).

Taką funkcjonalność można osiągnąć przez podłączenie książki adresowej CTI. Kilka przykładów podłączenia książek adresowych oferujemy pod adresem <http://www.ositron.de/de/ositron-cti-addins.html> w rubryce Książka adresow.

## 6 Dalsze kroki

Wychodząc z tej podstawowej integracji funkcji CTI z bazą danych Nordwind można teraz zrealizować inne pomysły.

Przykładowo można sobie wyobrazić, że przy wywołaniach przy- lub wychodzących nie będzie od razu wyświetlany rekord klienta, lecz najpierw pokazałby się ogólnie dostępny przycisk z krótko ujętymi informacjami klienta. Dopiero po naciśnięciu tego przycisku przez użytkownika, odpowiedni rekord klienta zostałby wyświetlony. Miałoby to tę zaletę, że użytkownik CTI i bazy Nordwind nie byłby odrywany od aktualnie wykonywanej pracy i mógłby wejrzeć w dane klienta później, najszybciej jak to byłoby możliwe.

Poza tym formularz klienta nie musiałby koniecznie być wcześniej otwarty, aby miała miejsce identyfikacja połączenia.

Na postawie zdarzeń sygnalizowanych dla połączeń telefonicznych można także prowadzić dalej idące działania dotyczące protokołowania danych klientów, tak że można przypisać połączenia do klientów, zleceń, projektów czy incydentów serwisowych w bazie danych i mieć do nich dostęp później.

Ponieważ integracja z OSITRON CTI może jednocześnie obsługiwać więcej połączeń, można by zrealizować inną funkcjonalność. Przykładowo można wyświetlić rekord dla połączeń pukających, podczas już prowadzonej rozmowy.

Użytkownik ma także możliwość wysyłania SMS za pośrednictwem integracji z CTI, zatem można sobie wyobrazić, że obok numeru telefonu komórkowego dodany zostałby dodatkowy przycisk, który otwierałby okienko wysyłania SMS wykorzystujące podany numer komórkowy.

Szczegółowe informacje można znaleźć w nocie technicznej dotyczącej integracji CTI, która znajduje się na naszym serwerze internetowym pod adresem <http://www.ositron.de/pl/com-integration.html>.